

Integrating Factors and Exact First Order ODE

Date: Jan 22, 2002

Last Revision: Jan 22, 2002

Maple 6

Bent E. Petersen

bent@alum.mit.edu

petersen@math.orst.edu

Course: Mth 256

Term: Winter 2002

File name: 256w2002-intfactors-exact-ode.mws

This worksheet demonstrates some of Maple's facilities for finding integrating factors and for solving exact ODEs.

The implicit solution of a first order exact ODE that we find by integration in Mth 256 is what is technically called a first integral. What we study in Mth 256 is a special case of a much more general concept. In Maple the command for finding first integrals of exact ODEs is `firint()`.

Before we can find a first integral we may first have to find an integrating factor (so we will have an exact equation to feed to `firint()`). Again the integrating factors studied in Mth 256 are a special case of a much more general concept. The Maple command for finding integrating factors is `intfactor()`.

Some of the problems below are from our text.

```
[ > restart;
```

We will need to load the DE tools library.

```
[ > with(DEtools);
```

Problem 1

First let's integrate an exact equation.

```
[ > ode01 := (3*x+2*y(x))*diff(y(x),x) + (2*x+3*y(x)) = 0;
```

$$ode01 := (3x + 2y(x)) \left(\frac{\partial}{\partial x} y(x) \right) + 2x + 3y(x) = 0$$

```
[ > soln01 := firint(ode01);
```

$$soln01 := 3xy(x) + y(x)^2 + x^2 + _C1 = 0$$

We can put the solution in a more customary form by making a couple of substitutions:

```
> subs(y(x)=y, _C1=C, soln01);
```

$$3xy + y^2 + x^2 + C = 0$$

I will make such cosmetic substitutions below without any additional comment.

Note the solution is quadratic in y . Thus if we try `dsolve()` it will return two explicit solutions here:

```
> s01:=dsolve(ode01);
```

$$s01 := y(x) = \frac{-\frac{3}{2}x_{_C1} - \frac{1}{2}\sqrt{5x^2_{_C1} + 4}}{_C1}, y(x) = \frac{-\frac{3}{2}x_{_C1} + \frac{1}{2}\sqrt{5x^2_{_C1} + 4}}{_C1}$$

```
> soln01a:=s01[1]: subs(_C1=C, y(x)=y, simplify(soln01a));
```

$$y = -\frac{1}{2} \frac{3xC + \sqrt{5x^2C^2 + 4}}{C}$$

```
> soln01b:=s01[2]: subs(_C1=C, y(x)=y, simplify(soln01b));
```

$$y = -\frac{1}{2} \frac{3xC - \sqrt{5x^2C^2 + 4}}{C}$$

Problem 2

Let's see how to find an integrating factor and then use it to solve a nonexact ODE.

```
> ode02 := (4*x*y(x)^2 + y(x)) + (6*y(x)^3 - x) * diff(y(x), x);
```

$$ode02 := 4xy(x)^2 + y(x) + (6y(x)^3 - x) \left(\frac{\partial}{\partial x} y(x) \right)$$

```
> mu02:=intfactor(ode02);
```

$$\mu02 := \frac{1}{y(x)^2}$$

```
> fint02:=firint(mu02*ode02): subs(_C1=C, y(x)=y, fint02);
```

$$3y^2 + \frac{x}{y} + 2x^2 + C = 0$$

Problem 3

Here's another nonexact ODE.

```
> ode03 := y(x) * (log(y(x)) + exp(x)) + (x + y(x) * cos(y(x))) * diff(y(x), x) = 0;
```

$$\text{ode03} := y(x) (\ln(y(x)) + e^x) + (x + y(x) \cos(y(x))) \left(\frac{\partial}{\partial x} y(x) \right) = 0$$

> `mu03:=intfactor(ode03);`

$$\mu03 := \frac{1}{y(x)}$$

> `firint(mu03*ode03): subs(_C1=C,y(x)=y,%);`

$$x \ln(y) + \sin(y) + e^x + C = 0$$

Problem 4

This time let's check for exactness before doing anything else.

> `M04:=7*x^4*y-3*y^8; N04:=2*x^5-9*x*y^7;`

$$M04 := 7x^4y - 3y^8$$

$$N04 := 2x^5 - 9xy^7$$

> `diff(M04,y)-diff(N04,x);`

$$-3x^4 - 15y^7$$

We see $M + N \, dy/dx$ is not exact, so let's look for an integrating factor and then integrate.

> `ode04:=subs(y=y(x),M04)+subs(y=y(x),N04)*diff(y(x),x)=0;`

$$\text{ode04} := 7x^4y(x) - 3y(x)^8 + (2x^5 - 9xy(x)^7) \left(\frac{\partial}{\partial x} y(x) \right) = 0$$

> `mu04:=intfactor(ode04);`

$$\mu04 := \frac{1}{x(y(x)^7 - x^4)y(x)}$$

> `firint(mu04*ode04): fir04:=subs(_C1=C,y(x)=y,%);`

$$\text{fir04} := 2 \ln(y) + \ln(y^7 - x^4) + 3 \ln(x) + C = 0$$

It looks like Maple missed an opportunity to simplify here. Let's exponentiate both sides of the equation fir04.

> `simplify(exp(lhs(fir04)))=exp(rhs(fir04)); subs(exp(C)=1/C,%);`

$$y^2 (y^7 - x^4) x^3 e^C = 1$$

$$\frac{y^2 (y^7 - x^4) x^3}{C} = 1$$

Here's another integrating factor (found by hand):

```
> mu04b:=x^2*y(x);
```

$$\mu_{04b} := x^2 y(x)$$

```
> firint(mu04b*ode04): subs(_C1=C,y(x)=y,%);
```

$$-x^7 y^2 + x^3 y^9 + C = 0$$

We have the same solution and we see that Maple does not always find the "simplest" integrating factor.!

Problem 5

```
> M05:=x^2*y^3; N05:=x*(1+y^2);
```

$$M_{05} := x^2 y^3$$

$$N_{05} := x(1 + y^2)$$

```
> ode05:=subs(y=y(x),M05)+subs(y=y(x),N05)*diff(y(x),x)=0;
```

$$\text{ode05} := x^2 y(x)^3 + x(1 + y(x)^2) \left(\frac{\partial}{\partial x} y(x) \right) = 0$$

```
> diff(M05,y)-diff(N05,x);
```

$$3x^2 y^2 - 1 - y^2$$

We see ODE 5 is not exact. Let's find an integrating factor and then a first integral.

```
> mu05:=intfactor(ode05);
```

$$\mu_{05} := \frac{1}{x y(x)^3}$$

```
> firint(mu05*ode05): subs(y(x)=y,_C1=C,%);
```

$$-\frac{1}{2} \frac{1}{y^2} + \ln(y) + \frac{1}{2} x^2 + C = 0$$

Problem 6

```
> M06:=sin(y)/y-2*exp(-x)*sin(x);
```

$$M_{06} := \frac{\sin(y)}{y} - 2 e^{(-x)} \sin(x)$$

```
> N06:=(cos(y)+2*exp(-x)*cos(x))/y;
```

$$N_{06} := \frac{\cos(y) + 2 e^{(-x)} \cos(x)}{y}$$

```
> ode06:=subs(y=y(x),M06)+subs(y=y(x),N06)*diff(y(x),x)=0;
```

$$ode06 := \frac{\sin(y(x))}{y(x)} - 2 e^{(-x)} \sin(x) + \frac{(\cos(y(x)) + 2 e^{(-x)} \cos(x)) \left(\frac{\partial}{\partial x} y(x) \right)}{y(x)} = 0$$

> `diff(M06,y) - diff(N06,x);`

$$\frac{\cos(y)}{y} - \frac{\sin(y)}{y^2} - \frac{-2 e^{(-x)} \cos(x) - 2 e^{(-x)} \sin(x)}{y}$$

We see ODE 6 is not exact. Let's look for an integrating factor.

> `mu06:=intfactor(ode06);`

$$\mu06 := \frac{1}{\sin(y(x)) + 2 e^{(-x)} \cos(x) y(x)}, \frac{e^x \cos(y(x)) + 2 \cos(x)}{\cos(y(x)) + 2 e^{(-x)} \cos(x)}$$

Maple returned two integrating factors:

> `mu06a:=mu06[1]; mu06b:=mu06[2];`

$$\mu06a := \frac{1}{\sin(y(x)) + 2 e^{(-x)} \cos(x) y(x)}$$

$$\mu06b := \frac{e^x \cos(y(x)) + 2 \cos(x)}{\cos(y(x)) + 2 e^{(-x)} \cos(x)}$$

> `firint(mu06a*ode06);`

Error, (in ODEtools/firint) The given ODE is not exact

> `firint(mu06b*ode06);`

Error, (in ODEtools/firint) The given ODE is not exact

Something is wrong! Maple 6 spent much time computing these two integrating factors, but they do not work. Maybe I made an error or maybe Maple needs help on this one. In any case we see that it is a good idea to check Maple's answers.

Fortunately an integrating factor is known for this problem (and it is a mystery that Maple does not find it).

> `mu06c:=y(x)*exp(x); # pulled out of the hat`

$$\mu06c := y(x) e^x$$

> `firint(mu06c*ode06): subs(y(x)=y, _C1=C, %);`

$$e^x \sin(y) + 2 e^x e^{(-x)} \cos(x) y + C = 0$$

Problem 7

```
> ode07 := (3*x+2*y(x)) + (4*x-2*y(x))*diff(y(x),x)=0;
```

$$ode07 := 3x + 2y(x) + (4x - 2y(x)) \left(\frac{\partial}{\partial x} y(x) \right) = 0$$

We can use firint() to determine if an ODE is exact (rather than checking by hand as above):

```
> firint(ode07);
```

```
Error, (in ODEtools/firint) The given ODE is not exact
```

```
> mu07:=intfactor(ode07);
```

$$\mu07 := \frac{1}{-3x^2 - 6xy(x) + 2y(x)^2}$$

```
> firint(mu07*ode07): subs(y(x)=y, _C1=C, %);
```

$$\frac{1}{4} \ln(-3x^2 - 6xy + 2y^2) + \frac{1}{30} \sqrt{15} \operatorname{arctanh} \left(\frac{1}{30} \frac{(-6x + 4y)\sqrt{15}}{x} \right) + C = 0$$

Problem 8

In this problem we have an ODE with a parameter and we want to know for what value of the parameter the equation is exact. Then solve the exact equation.

```
> M08:=y*exp(2*x*y)+x; N08:=b*x*exp(2*x*y);
```

$$M08 := y e^{(2xy)} + x$$

$$N08 := b x e^{(2xy)}$$

```
>
```

```
> ode08:=subs(y=y(x),M08)+subs(y=y(x),N08)*diff(y(x),x)=0;
```

$$ode08 := y(x) e^{(2xy(x))} + x + b x e^{(2xy(x))} \left(\frac{\partial}{\partial x} y(x) \right) = 0$$

```
> diff(M08,y)-diff(N08,x); simplify(exp(-2*x*y)*%); solve(%=0,b);
```

$$\frac{e^{(2xy)} + 2yx e^{(2xy)} - b e^{(2xy)} - 2bxy e^{(2xy)}}{1 + 2xy - b - 2bxy}$$

So b=1 makes ODE 8 exact:

```
> ode08b:=subs(b=1,ode08);
```

$$ode08b := y(x) e^{(2xy(x))} + x + x e^{(2xy(x))} \left(\frac{\partial}{\partial x} y(x) \right) = 0$$

```
> firint(ode08b): subs(y(x)=y, _C1=C, %);
```

$$\frac{1}{2} e^{(2xy)} + \frac{1}{2} x^2 + C = 0$$

Problem 9

> **M09:=3*x+6/y; N09:=x^2/y+3*y/x;**

$$M09 := 3x + \frac{6}{y}$$

$$N09 := \frac{x^2}{y} + \frac{3y}{x}$$

> **ode09:=subs(y=y(x),M09)+subs(y=y(x),N09)*diff(y(x),x)=0;**

$$ode09 := 3x + \frac{6}{y(x)} + \left(\frac{x^2}{y(x)} + \frac{3y(x)}{x} \right) \left(\frac{\partial}{\partial x} y(x) \right) = 0$$

> **mu09:=intfactor(ode09);**

$$\mu09 := x y(x), \frac{x y(x)}{3 x^2 + y(x)^3 + x^3 y(x)}$$

Note Maple returned two integrating factors here:

> **mu09a:=mu09[1]; mu09b:=mu09[2];**

$$\mu09a := x y(x)$$

$$\mu09b := \frac{x y(x)}{3 x^2 + y(x)^3 + x^3 y(x)}$$

> **firint(mu09a*ode09): subs(y(x)=y,_C1=C,%);**

$$3 x^2 + y^3 + x^3 y + C = 0$$

> **firint(mu09b*ode09): subs(y(x)=y,_C1=C,%);**

$$\ln(3 x^2 + y^3 + x^3 y) + C = 0$$

These two solutions are obviously "the same."

Problem 10

> **M10:=3*x^2*y+2*x*y+y^3; N10:=x^2+y^2;**

$$M10 := 3y x^2 + 2x y + y^3$$

$$N10 := x^2 + y^2$$

> **ode10:=subs(y=y(x),M10)+subs(y=y(x),N10)*diff(y(x),x)=0;**

$$ode10 := 3 x^2 y(x) + 2 x y(x) + y(x)^3 + (x^2 + y(x)^2) \left(\frac{\partial}{\partial x} y(x) \right) = 0$$

```
> mu10:=intfactor(ode10);
```

$$\mu_{10} := e^{(3x)}$$

```
> firint(mu10*ode10): subs(y(x)=y, _C1=C, %);
```

$$e^{(3x)} y x^2 + \frac{1}{3} e^{(3x)} y^3 + C = 0$$

Problem 11

Here's a linear equation

```
> ode11:=diff(y(x),x)=exp(2*x)+y(x)-1;
```

$$ode11 := \frac{\partial}{\partial x} y(x) = e^{(2x)} + y(x) - 1$$

```
> mu11:=intfactor(ode11);
```

$$\mu_{11} := e^{(-x)}$$

```
> firint(mu11*ode11): subs(y(x)=y, _C1=C, %);
```

$$e^{(-x)} y - e^x - e^{(-x)} + C = 0$$

We can of course also use dsolve() directly

```
> dsolve(ode11,y(x)): subs(y(x)=y, _C1=C, %);
```

$$y = e^{(2x)} + 1 + e^x C$$

Problem 12

```
> M12:=1; N12:=x/y-sin(y);
```

$$M12 := 1$$

$$N12 := \frac{x}{y} - \sin(y)$$

```
> ode12:=subs(y=y(x),M12)+subs(y=y(x),N12)*diff(y(x),x)=0;
```

$$ode12 := 1 + \left(\frac{x}{y(x)} - \sin(y(x)) \right) \left(\frac{\partial}{\partial x} y(x) \right) = 0$$

```
> mu12:=intfactor(ode12);
```

$$\mu_{12} := y(x)$$

```
> firint(mu12*ode12): subs(y(x)=y, _C1=C, %);
```

$$x y - \sin(y) + y \cos(y) + C = 0$$

Again dsolve() can handle this equation:

```
[ > dsolve(ode12,y(x)): subs(y(x)=y,_C1=C, %);
```

$$x - \frac{\sin(y) - y \cos(y) + C}{y} = 0$$

```
[ >
```